

MS 365 Excel Basics #4

Date Time Number Formatting, Formulas, Functions Calculations

Goal of video: Learn about date time serial numbers, number formatting formulas/functions

How Dates work in Excel (Date Serial Numbers, Date Number Format and Date Formulas)	2
Formulas to calculate number of days between two dates, for invoices, loans and projects	4
Formulas to count workdays with NETWORKDAYS.INTL function	5
Add days to date to get future date. Learn about WORKDAY.INTL function. Examples: calculating loan due dates	6
Subtract days from dates to get past date. Examples such as calculating the date that a project was started	6
Move date by a specified number of months with EDATE and EOMONTH functions. Examples with invoices	7
Extract the year, month or day from date using the functions, YEAR, MONTH, DAY, and TEXT. Examples for a date table ...	8
How Times work in Excel (Time Serial Numbers, Time Number Format and Time Formulas)	9
Time Formulas for Payroll	11
Time Formulas for Night Shift Payroll. Learn about Date-Time Values and the MOD function	12
List of All 26 Date & Time Formulas and Functions	13

	A	B	C	D	E	F	G	H	I	J	K	L
2		Invoice Due Date	10/25/2024		In USA enter dates like: Month/Day/Year or Month-Day-Year							
3		Today	11/20/2024		Keyboard for today's date: Ctrl + ;							
4		Days Past Due	26		Important Formula: Days between dates = Later Date - Earlier Date							
6		How does that work?	C4: =C3-C2									
8		Dates in Excel:										
9		When you type a date, an automatic Date Number Format is applied on top of a Date Serial Number,										
10		which is the number of days since Dec, 31, 1899. 1/1/1900 = 1, 1/2/1900 = 2										
12		Today	45616		General Number Format Keyboard = Ctrl + Shift + ~ / `							
14		Examples of Dates:										
16		Date	Serial Number									
17		1/1/1900	1		First day that Excel understands							
18		1/2/1900	2		Second day that Excel understands							
19		11/17/2024	45613		The date 11/17/2024 is 45613 days since Dec 31, 1899							
20		5/5/9999	2958225		The date 5/5/9999 is 2958225 days since Dec 31, 1899							
21		12/31/1899	Text		Text because it is before the first day that Excel recognizes as a date value.							
22		7/4/1776	Text		Text because it is before the first day that Excel recognizes as a date value.							
23		#####	Column not wide enough		#### tells you that the column not wide enough to display Date with Number Formatting							
25		There are Many Number Formats for Dates:										
26		Today	Serial Number									
27		11/17/2024	45613									
28		Sunday, November 17, 2024	45613									
29		11/17	45613									
30		17-Nov-24	45613									
31		Sunday, 11/17/2024	45613		Number Formatting Code: m = month, d = day, y = year							
32					Such as: ddd, m/d/yyyy to show Sun, 11/17/2024							

Date Custom Number Format :

d = day

m = month

y = year

Separators allowed: /, -

Date 11/1/2024

Date Custom Number Format	Date = 11/1/2024	What Number Format Shows
d	1	Day number
dd	01	Day number with lead zero
ddd	Fri	3 letter day
dddd	Friday	Full day name
m	11	Month number
mm	11	Month number with lead zero
mmm	Nov	3 letter month
mmmm	November	Full month name
y	24	Last 2 digits of year
yy	24	Last 2 digits of year
yyy	2024	4 digit year
yyyy	2024	4 digit year
m/d/yyy	11/1/2024	Forward slash ok
m-d-yyy	11-1-2024	Dash ok
ddd, m/d/yy	Fri, 11/1/24	Comma ok

	A	B	C	D	E	F	G	H	I
1									
2		Difference Between Two Dates							
3									
4		Question:	Start	End	Days:		Formula:	Notes:	
5		How many days until school starts?	8/15/2024	9/26/2024	42	1	E5: =D5-C5	Difference between dates, not including 1st day = EndDate - StartDate	
6		How many days is the invoice past due?	2/28/2024	3/1/2024	2	2	E6: =D6-C6		
7		How many days was the loan outstanding?	9/5/2024	1/3/2025	120	3	E7: =D7-C7		
8		How many days did the project take?	12/27/2024	3/1/2025	65	4	E8: =D8-C8+1	Difference between dates, including 1st day = EndDate - StartDate + 1	
9		How many days of vacation, including 1st day?	8/15/2024	9/26/2024	43	5	E9: =D9-C9+1		

	A	B	C	D	E	F	G	H	I	J
1										
2	Count Workdays (difference between dates, while skipping some days)									
3										
4	Question:	Start	End	Days:	Formula:			Holidays		
5	How many workdays are there? (Fri. & Sat. non-workdays)	1/21/2024	3/14/2024	38	6	E5: =NETWORKDAYS.INTL(C5,D5,7,J5:J8)			Mon, 1/15/2024	
6	How many workdays are there? (Wed. & Sun. non-workdays)	1/21/2024	3/14/2024	35	7	E6: =NETWORKDAYS.INTL(C6,D6,"0010001",J5:J8)			Mon, 1/22/2024	
7									Mon, 1/29/2024	
8									Fri, 2/16/2024	
9										
10										
11										
12										
13										
14										
15										

Notes:

=NETWORKDAYS.INTL(*StartDate* ,*EndDate* ,*WeekendString* ,*HolidaysAsRange*)

- * weekend argument you can create any pattern of workdays / non-workdays.
- * 1 represents a non-workday and 0 represents a workday.
- * Must be 7 characters long.
- * Must be encased in quotes.
- * For example, "0010001" would result in a weekend that is **Wednesday** and **Sunday**.

	A	B	C	D	E	F	G	H	I
1									
2		Add Days to Date to Get Future Date							
3									
4		Question:	Start	Number Days	Dates:		Formulas:	Notes:	
5		When is the loan due?	9/5/2024	120	1/3/2025		8 E5: =C5+D5	Calculate future date by adding days = StartDate + NumberOfDays	
6		What is end date for project (1st day IS included)? (Fri. & Sat. non-workdays)	11/17/2024	40	1/9/2025		9 E6: =WORKDAY.INTL(C6-1,D6,7)	Determine future date while excluding weekends and holidays (1st day included) = WORKDAY.INTL(StartDate-1,NumberWorkDays, WeekendNumber,HolidayAsRange)	
7									
8		Subtract Days from Date to Get Past Date							
9									
10		Question:	End	Number Days	Start		Formulas:	Notes:	
11		When was the loan issued?	1/3/2025	120	9/5/2024		10 E11: =C11-D11	Calculate future date by adding days = StartDate - NumberOfDays	
12		What is end date for project (1st day IS included)? (Fri. & Sat. non-workdays)	1/9/2025	-40	11/17/2024		11 E12: =WORKDAY.INTL(C12+1,D12,7)	Determine past date while excluding weekends and holidays (1st day included) = WORKDAY.INTL(StartDate+1,NumberWorkDays, WeekendNumber,HolidayAsRange)	

	A	B	C	D	E	F	G	H	I
1									
2		Move a Date a Number Months Forward or Backwards							
3									
4		Question:	Start Date	Number Months	Date:		Formulas:	Notes:	
5		When does my pension plan vest?	1/22/2024	60	1/22/2029	12	E5: =EDATE(C5,D5)	Calculate future date by adding number of months with: =EDATE(StartDate,NumberOfMonths)	
6		What is last day of this month?	2/20/2024	0	2/29/2024	13	E6: =EOMONTH(C6,D6)	Calculate end of month date with: =EOMONTH(StartDate,NumberOfMonths)	
7		What day is invoice due, if it is due on the same day, two months ahead?	1/22/2024	2	3/22/2024	14	E7: =EDATE(C7,D7)	For each function months can be: positive (jump into future), zero (same month as StartDate) negative (jump into past)	
8		What day is invoice due, if it is due at the end of next month?	1/20/2024	1	2/29/2024	15	E8: =EOMONTH(C8,D8)		

	A	B	C	D	E	F	G	H
1								
2	Year, Month, Day from Date Serial Number							
3								
4	Question:	Start Date	Date:				Formulas:	Notes:
5	Year	11/17/2024	2024	16	D5: =YEAR(C5)			YEAR creates the year as a number from a Serial Number Date
6	Month	11/17/2024	11	17	D6: =MONTH(C6)			MONTH creates the month as a number from a Serial Number Date
7	Day	11/17/2024	17	18	D7: =DAY(C7)			DAY creates the day as a number from a Serial Number Date
8	Day Name?	11/17/2024	Sunday	19	D8: =TEXT(C8,"dddd")			TEXT converts a number to text based on Number Formatting Code.
9	Month Name?	11/17/2024	Nov	20	D9: =TEXT(C9,"mmm")			

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1								Space						
2		Description	Time Value											
3		8:15:15 PM	8:15:15 PM											
4		20:15:15	20:15:15											
5		8:00 AM	8:00 AM											
6		5:15 PM	5:15 PM											
7		Current Time	6:48 PM											
8			11/21/2024 18:48											
9		Q: But what is actually in the cell, when you enter a time value?												
10		A: The proportion of a 24 hour day.												
11														
12		Type 8 a, Enter	8:00 AM											
13		Type 8 a, Ctrl + Enter, Ctrl + Shift + ~/`	0.333333333											
14		Type formula =8/24, Ctrl + Enter, Apply Time Number Format	8:05:00 AM											
15														
16		Q: How to calculate hours between two dates when Excel divides time value by 24?												
17		A: Multiply by 24 to get decimal back up to an hour unit, like: (End Time - Start Time)*24												
18														
19		Time In (Start)	8:00 AM											
20		Time Out (End)	5:15 PM											
21		Hours Worked	9.25											
22														
23														
24		Time In (Start1)	8:00 AM											
25		Time Out (End1)	12:05 PM											
26		Time In (Start2)	1:15 PM											
27		Time Out (End2)	5:15 PM											
28		Hours Worked	8.083333333											
29														
30														
31														
32		Time	Change Number Formatting											
33		8:00 PM	0.833333333											
34		08:00 PM	0.833333333											
35		08:00:00 PM	0.833333333											
36		20:00	0.833333333											

Enter times like: **hh:mm:ss AM/PM** , As soon as you type a colon, Excel thinks that it is a time value.
 Military time: don't type AM/PM
 Enter significant digits only for time values like: 8 a for 8 AM, and Excel understands it is a time value
 Enter significant digits only for time values like: 5 p for 5 PM, and Excel understands it is a time value
 Keyboard for today's time: Ctrl + Shift + ;

What Excel did behind the scenes when you entered 8:00 AM is: $8 / 24 = 1/3 = 0.333\dots$

The proportion of 8 hours to 24 hours in day with Time Number Formatting applied

(End Time - Start Time)*24 =(C20-C19)*24
 Remember: Ctrl + Shift + ~/` to remove Time Number Format

(EndTime1 - StartTime1+EndTime2 - StartTime2)*24 =(C25-C24+C27-C26)*24
 Do not round times: keep all the decimals.
 Remember: Ctrl + Shift + ~/` to remove Time Number Format

Number Formatting Code: **h** = hour, **m** = minute, **s** = second, separated by colon.
 Such as: **h:mm AM** to show **8:00 AM** or **h:mm** to show **20:00** for 8:00 PM

Time Custom Number Format :

h = hour

m = minute

s = seconds

: (colon) separates parts of time value

Type Space and then AM or PM

Leave out AM or PM, Military time is displayed

[hh] shows hours past 24 hours

Time with Time Number Format	Time Custom Number Format	Serial Number Under the Time
1:30 PM	h:mm AM/PM	0.5625
01:30 PM	hh:mm AM/PM	0.5625
1:30:00 PM	h:mm:ss AM/PM	0.5625
13:30	h:mm	0.5625
00:15:00	hh:mm:ss	0.010416667
00:00:15	hh:mm:ss	0.000173611
8 AM	h AM/PM	0.333333333
8/31/2022 8:03	m/d/yy h:mm	44804.33542
8/31/22 8:03 AM	m/d/yy h:mm AM/PM	44804.33542
30:00:00	[hh]:mm:ss	1.25
30:00	[hh]:mm	1.25
78:30	[hh]:mm	3.270833333

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2	Payroll Time Calculations											
3												
4	Employee	Hourly Wage (\$)	Start Time	End Time	Hours Worked	Gross Pay (\$)	* Assume we will use this formula result later					
5	Shardae Long	27.55	8:00 AM	1:10 PM	5.166666667	142.34	** 10/60 = 1/6 = 0.66... Time Value is not required to be rounded.					
6				Formulas:	F5: =(E5-D5)*24	G5: =ROUND(F5*C5,2)						
7				Do not round hours	Round Money							
8					21	22						
9												
10				Note:	↑ Number hours between two time values = (EndTime - StartTime)*24 Number hours when there is a lunch break = (EndTime1 - StartTime1 + EndTime2 - StartTime2)*24 ↓							
11												
12												
13												
14												
15												
16				Lunch Break								
17	Employee	Day	StartTime	EndTime1	StartTime2	EndTime2	Hours Worked	23 H17: =(E17:E21-D17:D21+G17:G21-F17:F21)*24				
18	Shardae Long	MONDAY	8:30 AM	12:00 PM	12:30 PM	5:42 PM	8.7					
19		TUESDAY	8:30 AM	12:00 PM	12:30 PM	5:00 PM	8					
20		WEDNESDAY	9:30 AM	12:10 PM	12:30 PM	4:00 PM	6.166666667					
21		THURSDAY	8:30 AM	12:00 PM	1:30 PM	5:30 PM	7.5					
22		FRIDAY	7:00 AM	10:30 AM	11:15 AM	3:00 PM	7.25					
23	TOTAL HOURS						Total Hours:	37.61666667	H22: =SUM(H17#)			
24							Wage per hour:	\$35.75				
							Gross Pay:	\$1,344.80	H24: =ROUND(H23*H22,2)			

Night Shift Time Calculations

Question:	Start	End	Hours Worked:
How many hours did you work on night shift?	11/21/2024 8:20 PM	11/22/2024 2:05 AM	5.75
How many hours did you work on night shift?	8:20 PM	2:05 AM	5.75
How many hours did you work on night shift?	8:20 PM	2:05 AM	5.75

Formula:	Notes:
24 E5: =(D5-C5)*24	If you are using date-time values, Hours Worked on Night Shift = (EndTime - StartTime)*24
24 E6: =MOD(D6-C6,1)*24	Hours Worked on Night Shift = MOD(EndTime - StartTime,1)*24
26 E7: =(D7-C7+(D7<C7))*24	Hours Worked on Night Shift = ((EndTime-StartTime + (EndTime<StartTime))*24

MOD(n, d) = n - d*INT(n/d)	
n	EndTime - StartTime
d	1
n - d*INT(n/d)	0.239583333

Formula:

C12: =D6-C6-C11*INT((D6-C6)/C11)

n - d*INT(n/d) = 2:05 AM - 8:20 PM - 1 * INT((2:05 AM - 8:20 PM)/1)

How Mod works:

<https://www.youtube.com/watch?v=VcgSkLpgo6o>

Excel Magic Trick 1206: Day & Night Shift Time Calculations: Add Total Hours Single Cell Formula

	A	B	C	D	E	F	G	H	I	J	K
1	Date and Time Formulas:										
2		Question:	Start	End	Days:		Formula:	Notes:			
3		How many days until school starts?	8/15/2024	9/26/2024	42		1 E3: =D3-C3	Difference between dates, not including 1st day = EndDate - StartDate			
4		How many days is the invoice past due?	2/28/2024	3/1/2024	2		2 E4: =D4-C4				
5		How many days was the loan outstanding?	9/5/2024	1/3/2025	120		3 E5: =D5-C5				
6		How many days did the project take?	12/27/2024	3/1/2025	65		4 E6: =D6-C6+1	Difference between dates, including 1st day = EndDate - StartDate + 1			
7		How many days of vacation, including 1st day?	8/15/2024	9/26/2024	43		5 E7: =D7-C7+1				
8		How many workdays are there? (Fri. & Sat. non-workdays)	1/21/2024	3/14/2024	38		6 E8: =NETWORKDAYS.INTL(C8,D8,7,K6:K9)	How many Working Days = NETWORKDAYS.INTL(StartDate,EndDate,WeekendNumber,HolidaysAsRange)			
9		How many workdays are there? (Wed. & Sun. non-workdays)	1/21/2024	3/14/2024	35		7 E9: =NETWORKDAYS.INTL(C9,D9,"0010001",K6:K9)				
10											
11		Question:	Start	Number Days	Dates:		Formulas:	Notes:			
12		When is the loan due?	9/5/2024	120	1/3/2025		8 E12: =C12+D12	Calculate future date by adding days = StartDate + NumberOfDays			
13		What is end date for project (1st day IS included)? (Fri. & Sat. non-workdays)	11/17/2024	40	1/9/2025		9 E13: =WORKDAY.INTL(C13-1,D13,7)	Determine future date while excluding weekends and holidays (1st day included) = WORKDAY.INTL(StartDate-1,NumberOfDays,WeekendNumber,HolidayAsRange)			
14											
15		Question:	End	Number Days	Start		Formulas:	Notes:			
16		When was the loan issued?	1/3/2025	120	9/5/2024		10 E16: =C16-D16	Calculate future date by adding days = StartDate - NumberOfDays			
17		What is end date for project (1st day IS included)? (Fri. & Sat. non-workdays)	1/9/2025	-40	11/17/2024		11 E17: =WORKDAY.INTL(C17+1,D17,7)	Determine past date while excluding weekends and holidays (1st day included) = WORKDAY.INTL(StartDate+1,NumberOfDays,WeekendNumber,HolidayAsRange)			
18											

Holidays
Mon, 1/15/2024
Mon, 1/22/2024
Mon, 1/29/2024
Fri, 2/16/2024

	A	B	C	D	E	F	G	H	I	J	K
1		Date and Time Formulas:									
19		Question:	Start Date	Number Months	Date:			Formulas:	Notes:		
20		When does my pension plan vest?	1/22/2024	60	1/22/2029			12 E20: =EDATE(C20,D20)	Calculate future date by adding number of months with: =EDATE(StartDate,NumberOfMonths)		
21		What is last day of this month?	2/20/2024	0	2/29/2024			13 E21: =EOMONTH(C21,D21)	Calculate end of month date with: =EOMONTH(StartDate,NumberOfMonths)		
22		What day is invoice due, if it is due on the same day, two months ahead?	1/22/2024	2	3/22/2024			14 E22: =EDATE(C22,D22)	For each function months can be: positive (jump into future), zero (same month as StartDate) negative (jump into past)		
23		What day is invoice due, if it is due at the end of next month?	1/20/2024	1	2/29/2024			15 E23: =EOMONTH(C23,D23)			
24											
25		Question:	Start Date		Date:						
26		Year	11/17/2024		2024			16 E26: =YEAR(C26)	YEAR creates the year as a number from a Serial Number Date		
27		Month	11/17/2024		11			17 E27: =MONTH(C27)	MONTH creates the month as a number from a Serial Number Date		
28		Day	11/17/2024		17			18 E28: =DAY(C28)	DAY creates the day as a number from a Serial Number Date		
29		Day Name?	11/17/2024		Sunday			19 E29: =TEXT(C29,"dddd")	TEXT converts a number to text based on Number		
30		Month Name?	11/17/2024		Nov			20 E30: =TEXT(C30,"mmm")	Formatting Code.		
31											
32		Question:		Start Time	8:00 AM						
33		Calculate Gross Pay		End Time	1:10 PM						
34				Hours Worked	5.166666667			21 E34: =(E33-E32)*24	Number hours between two time values =(EndTime - StartTime)*24 . Do not round hours.		
35				Hour Wage (\$)	27.55						
36				Gross Pay (\$)	142.34			22 E36: =ROUND(E34*E35,2)	For payroll, you need to round to the penny.		
37											
38		Question:		StartTime	8:30 AM						
39		Calculate Gross Pay with Lunch Break:		EndTime1	12:05 PM						
40				StartTime2	12:30 PM						
41				EndTime2	5:42 PM						
42				Hours Worked	8.783333333			23 E42: =(E39-E38+E41-E40)*24	Number hours when there is a lunch break =(EndTime1 - StartTime1 + EndTime2 - StartTime2)*24 . Do not round hours.		
43											
44		Question:	Start	End	Hours Worked:			Formula:	Notes:		
45		How many hours did you work on night shift?	11/21/2024 8:20 PM	11/22/2024 2:05 AM	5.75			24 E45: =(D45-C45)*24	If you are using date-time values, Hours Worked on Night Shift = (EndTime - StartTime)*24		
46		How many hours did you work on night shift?	8:20 PM	2:05 AM	5.75			24 E46: =MOD(D46-C46,1)*24	Hours Worked on Night Shift = MOD(EndTime - StartTime,1)*24		
47		How many hours did you work on night shift?	8:20 PM	2:05 AM	5.75			26 E47: =(D47-C47+(D47<C47))*24	Hours Worked on Night Shift = ((EndTime-StartTime + (EndTime<StartTime))*24		